Chapter 9
Topics in Transmission Operation and Planning

This chapter provides a solution for some transmission network operation and planning studies in GAMS. The transmission investment regarding building new lines and power flow controllers (phase shifter), sensitivity factors, and transmission switching have been discussed in this chapter. The GAMS code for solving each optimization problem is developed and discussed.

9.1 Transmission Network Planning

The question to be answered in transmission expansion planning (TEP) is when and which right of way should be selected to build a new line, perform reconductoring, build a new substation, or install power flow controllers. As a matter of fact, obtaining the public acceptance for building new transmission lines has become a challenging issue for the transmission asset owners. This also makes it difficult for transmission system operator to keep the technical and economic performance of transmission network high. The objective function is usually defined as the total operation and planning costs. Different models have been proposed for TEP purpose such as:

- Probabilistic TEP considering load and wind power generation uncertainties [1]
- MIP-based multi-stage TEP considering losses, generator costs, and the $N-1$ security constraints [2]
- Genetic algorithm-based TEP considering demand uncertainty [3]
- Congestion reduction-based TEP [4]
- Robust optimization-based TEP considering the uncertainties of renewable generation and load [5]
- Branch and bound algorithm for TEP [6]
- MIP-based TEP model considering different demand levels, $N-1$ network security constraints as well as environmental constraints [7]

A. Soroudi, *Power System Optimization Modeling in GAMS*,
DOI 10.1007/978-3-319-62350-4_9

- Multi-objective TEP considering total social costs, maximum regret (robustness criterion), and maximum adjustment cost (flexibility criterion) as three objective functions [8]
- Monte-Carlo-based TEP considering random outages of generating units and transmission lines as well as inaccuracies in the long-term load forecasting [9]
- An interior point method considering full AC power flow constraints [10]
- Multi-objective TEP considering investment cost, reliability (both adequacy and security), and congestion costs [11]
- Chance constrained TEP consideration of load and wind uncertainties [12].

9.1.1 TEP with New Lines Option

The transmission expansion planning is formulated in (9.1). It is assumed that the only planning option is building new lines.

$$\mathrm{OF} = T \times \mathrm{OPC} + \mathrm{INVC} \tag{9.1a}$$

$$\mathrm{OPC} = \sum_{g \in \Omega_G} b_g P_g + \mathrm{VOLL} \sum_i \mathrm{LS}_i \tag{9.1b}$$

$$\mathrm{INVC} = \left(-\eta_{ij}^0 + \sum_{k,ij} \alpha_{ij}^k \right) C_{ij} \tag{9.1c}$$

$$P_{ij}^k - B_{ij}(\delta_i - \delta_j) \le (1 - \alpha_{ij}^k) M \tag{9.1d}$$

$$P_{ij}^k - B_{ij}(\delta_i - \delta_j) \ge -(1 - \alpha_{ij}^k) M \tag{9.1e}$$

$$\sum_{g \in \Omega_G^i} P_g + \mathrm{LS}_i - L_i = \sum_{j \in \Omega_\ell^i} P_{ij} : \lambda_i \quad i \in \Omega_B \tag{9.1f}$$

$$-P_{ij}^{\max} \alpha_{ij}^k \le P_{ij}^k \le P_{ij}^{\max} \alpha_{ij}^k \quad ij \in \Omega_\ell \tag{9.1g}$$

$$P_g^{\min} \le P_g \le P_g^{\max} \tag{9.1h}$$

$$\text{if } \eta_{ij}^0 = 1 \text{ then } \alpha_{ij}^{k=1} = 1 \tag{9.1i}$$

$$B_{ij} = \frac{1}{x_{ij}} \tag{9.1j}$$

$$\alpha_{ij}^k \in \{0, 1\} \tag{9.1k}$$

$$k \in \{1, 2, 3, 4\} \tag{9.1l}$$



Fig. 9.1 Six-bus Garver transmission network (base case)

The objective function in (9.1a) consists of operational costs (OPC) and investment costs (INVC). The operational costs are calculated in (9.1b) while the investment costs are calculated in (9.1c). (9.1d), and (9.1e) model, the power flow on branch connecting bus i to bus j. In (9.1d) and (9.1e) there is a parameter M. It is also called big M in the literature [13]. This parameter is selected as follows:

$$M = \max_{ij} B_{ij}(\delta_i - \delta_j)$$

The power balance between the generated power, load shedding, demand, and line flows is ensured by (9.1f). The line flow limits are modeled in (9.1g) and the impacts of line investment decision α_{ij}^k are formulated. The generation operating limits are given in (9.1h). The initial status of each line is described in (9.1i).

The six-bus Garver transmission network [14] is shown in Fig. 9.1.

Transmission expansion planning data are given in Table 9.1. The existing and potential right of ways in addition to the reactances and flow limits are provided there. The investment costs (C_{ij}) are given in million \$. The VOLL is assumed to be 1000 \$/MW h. There are three generating units with different operating costs as shown in Fig. 9.1. Although the generator number 3 is the cheapest unit, however, it is not connected to the grid. The candidate right of ways are the existing ones (indicated by solid lines and the dashed ones as indicated in Fig. 9.1). The GAMS code for solving the DC power flow-based TEP is provided in GCode 9.1.

Table 9.1 Transmission expansion planning data for Garver six-bus transmission network

From (i)	To (j)	x_{ij}	$\bar{f}_{ij}$	C_{ij}	η^0_{ij}
1	2	0.40	100	40	1
1	4	0.60	80	60	1
1	5	0.20	100	20	1
2	3	0.20	100	20	1
2	4	0.40	100	40	1
2	6	0.30	100	30	0
3	5	0.20	100	20	1
4	6	0.30	100	30	0

GCode 9.1 DC-OPF-based TEP

```
Sets bus  /1*6/, slack(bus) /1/, Gen /g1*g3/, k /k1*k4/;
Scalars  Sbase /100/  , M /1000/; alias(bus,node);
Table GenData(Gen,*)  Generating units characteristics
   b      pmin pmax
g1 20     0      400
g2 30     0      400
g3 10     0      600;
set GBconect(bus,Gen) connectivity index of each generating unit to each bus
/1     .     g1
 3     .     g2
 6     .     g3  / ;
Table BusData(bus,*) Demands of each bus in MW
          Pd
1         80
2         240
3         40
4         160
5         240;
table branch(bus, node,*)     Network technical characteristics
                X     LIMIT Cost stat
1     .     2    0.4   100    40    1
1     .     4    0.6   80     60    1
1     .     5    0.2   100    20    1
2     .     3    0.2   100    20    1
2     .     4    0.4   100    40    1
2     .     6    0.3   100    30    0
3     .     5    0.2   100    20    1
4     .     6    0.3   100    30    0;
Set conex(bus,node)          Bus connectivity matrix;
conex(bus,node)$(branch(bus,node,'x'))=yes; conex(bus,node)$conex(node,bus)=yes;
branch(bus,node,'x')$branch(node,bus,'x')=branch(node,bus,'x');
branch(bus,node,'cost')$branch(node,bus,'cost')=branch(node,bus,'cost');
branch(bus,node,'stat')$branch(node,bus,'stat')=branch(node,bus,'stat');
branch(bus,node,'Limit')$(branch(bus,node,'Limit')=0)=branch(node,bus,'Limit');
branch(bus,node,'bij')$conex(bus,node) =1/branch(bus,node,'x');
M=smax((bus,node)$conex(bus,node),branch(bus,node,'bij')*pi*4/3);
Variables  OF, Pij(bus,node,k),Pg(Gen),delta(bus),LS(bus);
binary variable alpha(bus,node,k); alpha.l(bus,node,k)=1;
alpha.fx(bus,node,k)$(conex(bus,node) and ord(k)=1 and  branch(node,bus,'stat') )
     =1;
```

```
Equations const1A,const1B,const1C,const1D,const1E,const2,const3;
const1A(bus,node,k)$conex(node,bus) .. Pij(bus,node,k)-
branch(bus,node,'bij')*(delta(bus)-delta(node))=l= M*(1-alpha(bus,node,k));
const1B(bus,node,k)$conex(node,bus) .. Pij(bus,node,k)-
branch(bus,node,'bij')*(delta(bus)-delta(node))=g=-M*(1-alpha(bus,node,k));
const1C(bus,node,k)$conex(node,bus) .. Pij(bus,node,k)=l=
 alpha(bus,node,k)*branch(bus,node,'Limit')/Sbase;
const1D(bus,node,k)$conex(node,bus) .. Pij(bus,node,k)=g=
-alpha(bus,node,k)*branch(bus,node,'Limit')/Sbase;
const1E(bus,node,k)$conex(node,bus) .. alpha(bus,node,k)=e=alpha(node,bus,k);
const2(bus) .. LS(bus)+sum(Gen$GBconect(bus,Gen),Pg(Gen))-BusData(bus,'pd')/Sbase
=e=+sum((k,node)$conex(node,bus),Pij(bus,node,k));
const3 .. OF=g=2*8760*(sum(Gen,Pg(Gen)*GenData(Gen,'b')*Sbase)
+1000*sbase*sum(bus,LS(bus)) )
+1e6*sum((bus,node,k)$conex(node,bus),0.5*branch(bus,node,'cost')*alpha(bus,node,k)
$(ord(k)>1 or branch(node,bus,'stat')=0));
Model loadflow /all/; option optcr=0;
LS.up(bus)=BusData(bus,'pd')/Sbase; LS.lo(bus)=0;
Pg.lo(Gen)=GenData(Gen,'Pmin')/Sbase;
Pg.up(Gen)=GenData(Gen,'Pmax')/Sbase;
delta.up(bus)=pi/3; delta.lo(bus)=-pi/3; delta.fx(slack)=0;
Pij.up(bus,node,k)$((conex(bus,node)))=1*branch(bus,node,'Limit')/Sbase;
Pij.lo(bus,node,k)$((conex(bus,node)))=-1*branch(bus,node,'Limit')/Sbase;
Solve loadflow min OF us MIP;
```

9.1.1.1 Two Years Planning Period

The optimal TEP solution for 2 years planning period ($T = 2 * 8760\,\mathrm{h}$) is shown in Fig. 9.2. The total operating cost is M\$227.142, and the investment costs are M\$110. The total costs would be M\$337.140.

The optimal solutions dictate that the branch connecting bus 1 to bus 5 should be reinforced with one additional line. Bus 6 to bus 2 should be connected using three lines.

9.1.1.2 Ten Years Planning Period

The optimal TEP solution for 2 years planning period ($T = 10 * 8760\,\mathrm{h}$) is shown in Fig. 9.3. The total operating cost is M\$871.410 and the investment costs are M\$162.853. The total costs would be M\$1034.263. The optimal solutions dictates that the branch_{2-3} and branch_{1-5} should be reinforced with one additional line. Bus 6 to bus 2 should be connected using four lines. Bus 6 to bus 4 should be connected using two lines. It should be noted that the current model is simplistic. It needs to consider more realistic constraints. Some of them are listed as follows:



Fig. 9.2 Optimal TEP solution for 2 years planning period ($T = 2 * 8760\,\mathrm{h}$)



Fig. 9.3 Optimal TEP solution for 10 years planning period ($T = 10 * 8760\,\mathrm{h}$)

- The model is a single period. The load duration curve (LDC) or a discrete LDC with some demand levels and their associated duration should be considered.
- The contingencies should be taken into account. This will ensure the system operator to keep the light on even some network/generation assets fail. The failure of transmission lines or generating units might cause overloading of the remaining lines and operation of over-current relays and cause cascaded failures. It might even lead to black out.
- The AC power flow constraints should be checked to make sure no voltage constraint or line limit is violated.
- The voltage stability issue should also be checked.
- The model is trying to minimize the total operating cost plus the investment costs. In deregulated environment, the transmission asset owner and the transmission system operator are two independent entities. This makes the problem more complicated since these two entities would have different objective functions. The asset owner is trying to maximize its benefit and make money by making investments. On the other hand, the system operator is trying to maximize the social welfare. The multi-objective techniques [8] or complementarity models [15] provide the suitable answer to this challenge.
- The demand grows should be considered.
- The formulated transmission planning model is a static model. This means that the decision is made in order to make the system capable of answering for the needs of next N years. The dynamic of investment or timing of investment and time value of money is neglected.
- The uncertainties of the electricity market, renewable energies, demand, and regulatory frameworks should be considered to make the model robust against future scenarios.
- The VOLL is assumed to be the same for all demands at different buses. However, the importance level of all demands is not the same. This can reduce the investment requirements for TEP.

9.1.2 TEP with New Lines and Power Flow Controller Option

This section investigates the impact of power flow controller as a planning option. The phase shifter is used as the power flow controller device. The role of the phase shifter is depicted in Fig. 9.4. The relation between the power flow, voltage angles difference across the branch$_{ij}$, and the line susceptance is given in (9.2).

$$P_{ij} = \frac{\delta_i - \delta_j + \Psi_{ij}}{x_{ij}} = B_{ij}(\delta_i - \delta_j + \Psi_{ij}) \tag{9.2a}$$

$$-\Psi_{ij}^{\max} \le \Psi_{ij} \le +\Psi_{ij}^{\max} \tag{9.2b}$$

$$\Psi_{ij} = -\Psi_{ji} \tag{9.2c}$$



Fig. 9.4 The phase shifter function in power flow control

Although the voltage phase shift is a discrete variable in reality but it is modeled as a continuous variable for simplicity in (9.2).

$$\text{OF} = T \times \text{OPC} + \text{INVC} \tag{9.3a}$$

$$\text{OPC} = \sum_{g \in \Omega_G} b_g P_g + \text{VOLL} \sum_i \text{LS}_i \tag{9.3b}$$

$$\text{INVC} = \left(-\eta_{ij}^0 + \sum_{k,ij} \alpha_{ij}^k \right) C_{ij} + \sum_{k,ij} I_{ij}^k \gamma_{ij} \tag{9.3c}$$

$$P_{ij}^k - B_{ij}(\delta_i - \delta_j + \Psi_{ij}^k) \leq (1 - \alpha_{ij}^k)M \tag{9.3d}$$

$$P_{ij}^k - B_{ij}(\delta_i - \delta_j + \Psi_{ij}^k) \geq -(1 - \alpha_{ij}^k)M \tag{9.3e}$$

$$-\Psi_{ij}^{\max} I_{ij}^k \leq \Psi_{ij}^k \leq +\Psi_{ij}^{\max} I_{ij}^k \tag{9.3f}$$

$$\Psi_{ij} = -\Psi_{ji} \tag{9.3g}$$

$$I_{ij}^k \leq \alpha_{ij}^k \tag{9.3h}$$

$$\sum_{g \in \Omega_G^i} P_g + \text{LS}_i - L_i = \sum_{j \in \Omega_\ell^i} P_{ij} : \lambda_i \quad i \in \Omega_B \tag{9.3i}$$

$$-P_{ij}^{\max} \alpha_{ij}^k \leq P_{ij}^k \leq P_{ij}^{\max} \alpha_{ij}^k \quad ij \in \Omega_\ell \tag{9.3j}$$

$$P_g^{\min} \leq P_g \leq P_g^{\max} \tag{9.3k}$$

$$\text{if } \eta_{ij}^0 = 1 \text{ then } \alpha_{ij}^{k=1} = 1 \tag{9.3l}$$

$$B_{ij} = \frac{1}{x_{ij}} \tag{9.3m}$$

$$\alpha_{ij}^k, I_{ij}^k \in \{0, 1\} \tag{9.3n}$$

$$k \in \{1, 2, 3, 4\} \tag{9.3o}$$



Fig. 9.5 The optimal decisions regarding the location of phase shifter and new transmission lines 2 years plan ($T = 2 * 8760$)

The phase shifter impact is shown in (9.3d) and (9.3e). The investment decision regarding the phase shifter is reflected in (9.3f). The phase shifter can be installed on lines that initially exist or built lines as shown in (9.3h). The investment cost for each phase shifter is assumed to be $\gamma_{ij} = \text{M\$0.6}$. The optimal decisions regarding the location of phase shifter and new transmission lines 2 years plan ($T = 2 * 8760$) are given in Fig. 9.5. The total operating cost is M$224.256 and the investment costs are M$110.6. The total costs would be M$334.856.

GCode 9.2 DC-OPF-based TEP considering phase shifter option

```
Sets bus  /1*6/  ,slack(bus) /1/,Gen /g1*g3/,k /k1*k4/; alias(bus,node);
Scalars  Sbase /100/ , M /1000/, T; T=8760*2; Set conex(bus,node);
Table GenData(Gen,*)  Generating units characteristics
   b      pmin  pmax
g1  20     0     400
g2  30     0     400
g3  10     0     600;
Set GBconect(bus,Gen) connectivity index of each generating unit to each bus
/1     .     g1
 3     .     g2
 6     .     g3   / ;
Table BusData(bus,*) Demands of each bus in MW
          Pd
1         80
2         240
```

```
3           40
4           160
5           240;
table branch(bus, node,*)    Network technical characteristics
                 X      LIMIT  Cost  stat
;
conex(bus,node)$(branch(bus,node,'x'))=yes; conex(bus,node)$conex(node,bus)=yes;
branch(bus,node,'x')$branch(node,bus,'x')=branch(node,bus,'x');
branch(bus,node,'cost')$branch(node,bus,'cost')=branch(node,bus,'cost');
branch(bus,node,'stat')$branch(node,bus,'stat')=branch(node,bus,'stat');
branch(bus,node,'Limit')$(branch(bus,node,'Limit')=0)=branch(node,bus,'Limit');
branch(bus,node,'bij')$conex(bus,node) =1/branch(bus,node,'x');
M=smax((bus,node)$conex(bus,node),branch(bus,node,'bij')*3.14*2);
Variables  OF, Pij(bus,node,k),Pg(Gen),delta(bus),LS(bus),PSHij(bus,node,k);
binary variable alpha(bus,node,k),I(bus,node,k); alpha.l(bus,node,k)=1;
alpha.fx(bus,node,k)$(conex(bus,node) and ord(k)=1 and  branch(node,bus,'stat') )
      =1;
Equations const1A,const1B,const1C,const1D,const1E,const2,const3,
const4,const5,const6,const7;
const1A(bus,node,k)$conex(node,bus) .. Pij(bus,node,k)-branch(bus,node,'bij')*(
                  delta(bus)-delta(node)+PSHij(bus,node,k))=l=M*(1-alpha(bus,node,k))
                        ;
const1B(bus,node,k)$conex(node,bus) .. Pij(bus,node,k)-branch(bus,node,'bij')*(
                 delta(bus)-delta(node)+PSHij(bus,node,k))=g=-M*(1-alpha(bus,node,k))
                       ;
const1C(bus,node,k)$conex(node,bus) .. Pij(bus,node,k)=l=
alpha(bus,node,k)*branch(bus,node,'Limit')/Sbase;
const1D(bus,node,k)$conex(node,bus) .. Pij(bus,node,k)=g=
-alpha(bus,node,k)*branch(bus,node,'Limit')/Sbase;
const1E(bus,node,k)$conex(node,bus) .. alpha(bus,node,k)=e=alpha(node,bus,k);
const2(bus) .. LS(bus)+sum(Gen$GBconect(bus,Gen),Pg(Gen))
-BusData(bus,'pd')/Sbase=e=+sum((k,node)$conex(node,bus),Pij(bus,node,k));
const3 ..OF=g=T*(sum(Gen,Pg(Gen)*GenData(Gen,'b')*Sbase)+100000*sum(bus,LS(bus)))
+1e6*sum((bus,node,k)$conex(node,bus),0.5*branch(bus,node,'cost')*
alpha(bus,node,k)$(ord(k)>1 or branch(node,bus,'stat')=0))+6e5*0.5*sum((bus,node,k)
$conex(node,bus),I(bus,node,k));
const4(bus,node,k)$conex(node,bus) ..PSHij(bus,node,k)+PSHij(node,bus,k)=e=0;
const5(bus,node,k)$conex(node,bus) ..PSHij(bus,node,k)=l=I(bus,node,k)*pi/8;
const6(bus,node,k)$conex(node,bus) ..PSHij(bus,node,k)=g=-I(bus,node,k)*pi/8;
const7(bus,node,k)$conex(node,bus) ..I(bus,node,k)=l=alpha(bus,node,k);
Model loadflow     /all/; LS.up(bus)=BusData(bus,'pd')/Sbase; LS.lo(bus)=0;
Pg.lo(Gen)=GenData(Gen,'Pmin')/Sbase; Pg.up(Gen)=GenData(Gen,'Pmax')/Sbase;
delta.up(bus)=pi/3; delta.lo(bus)=-pi/3; delta.fx(slack)=0;
Pij.up(bus,node,k)$((conex(bus,node)))=1*branch(bus,node,'Limit')/Sbase;
Pij.lo(bus,node,k)$((conex(bus,node)))=-1*branch(bus,node,'Limit')/Sbase;
PSHij.up(bus,node,k)= pi/8; PSHij.lo(bus,node,k)=-pi/8; option optcr=0;
Solve loadflow min OF us MIP;
```

The optimal decisions regarding the location of phase shifter and new transmission lines 10 years plan ($T = 10 * 8760$) are given in Fig. 9.6. The total operating cost is M\$805.920 and the investment costs are M\$200.600. The total costs would be M\$1006.520.



Fig. 9.6 The optimal decisions regarding the location of phase shifter and new transmission lines 10 years plan ($T = 10 * 8760$)

Nomenclature

Indices and Sets

g	Index of thermal generating units
i, j	Index of network buses
Ω_G	Set of all thermal generating units
Ω_G^i	Set of all thermal generating units connected to bus i
Ω_ℓ^i	Set of all buses connected to bus i
Ω_B	Set of network buses

Parameters

M	Big number
T	Duration of planning period (h)
L_i	Electric power demand in bus i at time t
b_g	Fuel cost coefficient of thermal unit g
η_{ij}^0	Initial status of branch connecting bus i to j
C_{ij}	Investment cost for branch connecting bus i to j
γ_{ij}	Investment cost of phase shifter in line ij
$P_g^{\max/\min}$	Maximum/minimum limits of power generation of thermal unit g

$P_{ij}^{\max}$	Maximum power flow limits of branch connecting bus i to j
$\Psi_{ij}^{\max}$	Maximum phase shift in line connecting bus i to bus j
x_{ij}	Reactance of branch connecting bus i to j
B_{ij}	Susceptance of branch connecting bus i to j
VOLL	Value of loss of load ($/MW h)

Variables

P_{ij}^{k}	Active power flow of branch k connecting bus i to j (MW)
P_g	Active power generated by thermal unit g (MW)
α_{ij}^{k}	Binary variable to model the investment decision regarding the line k at the right of way ij
I_{ij}^{k}	Binary variable to model the investment decision regarding the phase shifter in line k at the right of way ij
λ_i	Locational marginal price in bus i ($/MW h)
LS_i	Load shedding in bus i (MW)
Ψ_{ij}	Phase shift in line connecting bus i to bus j
OPC	Total operating costs ($)
OF	Total costs ($)
INVC	Total investment costs ($)
δ_i	Voltage angle of bus i (rad)

9.2 Sensitivity Factors in Transmission Networks

In this section two important factors are analyzed and calculated namely:

- Generation Shift Factor (GSF)
- Line Outage Distribution Factor (LODF)

9.2.1 Generation Shift Factors

The transmission network planner/operator is always interested to know what happens to the line flows if any outage happens in generation units. In other words, what is the influence of generation/demand change in bus i on the line ℓ (connecting bus n to bus m)? It is important since there is always a chance for contingencies to happen in generating units. This section will provide an answer to this question.

The DC power flow equations of the network shown in Fig. 9.7 is provided in (9.4).

$$P_1^g - L_1 = \frac{\delta_1 - \delta_2}{x_{12}} + \frac{\delta_1 - \delta_3}{x_{13}} \tag{9.4a}$$



Fig. 9.7 Three-bus network example for sensitivity factors calculation

$$P_2^g - L_2 = \frac{\delta_2 - \delta_1}{x_{21}} + \frac{\delta_2 - \delta_3}{x_{23}} \tag{9.4b}$$

$$P_3^g - L_3 = \frac{\delta_3 - \delta_2}{x_{32}} + \frac{\delta_3 - \delta_1}{x_{31}} \tag{9.4c}$$

By substituting the numerical values of x_{ij} in (9.4) we will have :

$$P_1^g - L_1 = \frac{\delta_1 - \delta_2}{0.1} + \frac{\delta_1 - \delta_3}{0.2} \tag{9.5a}$$

$$P_2^g - L_2 = \frac{\delta_2 - \delta_1}{0.1} + \frac{\delta_2 - \delta_3}{0.25} \tag{9.5b}$$

$$P_3^g - L_3 = \frac{\delta_3 - \delta_2}{0.25} + \frac{\delta_3 - \delta_1}{0.2} \tag{9.5c}$$

(9.4) can be written as : $\begin{pmatrix} P_1 \\ P_2 \\ P_3 \end{pmatrix} = \begin{pmatrix} 15 & -10 & -5 \\ -10 & 14 & -4 \\ -5 & -4 & 9 \end{pmatrix} = \begin{pmatrix} \delta_1 \\ \delta_2 \\ \delta_3 \end{pmatrix}$ where $P_i = P_i^g - L_i$.

Suppose $B = \begin{pmatrix} 15 & -10 & -5 \\ -10 & 14 & -4 \\ -5 & -4 & 9 \end{pmatrix}$ then we can have the relation between the bus angles and the net injections as a linear matrix form.

$$P = B\delta \tag{9.6}$$

In normal DC power flow, the P vector is known and the decision maker's goal is to find the δ vector. The problem is that matrix B is singular and does not have a matrix inverse.

The good news is that the bus angle at slack bus is known to be zero. If the slack bus is bus 1 then we can have the following matrix form:

$$\begin{pmatrix} P_2 \\ P_3 \end{pmatrix} = \begin{pmatrix} 14 & -4 \\ -4 & 9 \end{pmatrix} \begin{pmatrix} \delta_2 \\ \delta_3 \end{pmatrix} \tag{9.7}$$

Now the matrix can be inversed as follows

$$\begin{pmatrix} \delta_2 \\ \delta_3 \end{pmatrix} = \begin{pmatrix} 14 & -4 \\ -4 & 9 \end{pmatrix}^{-1} \begin{pmatrix} P_2 \\ P_3 \end{pmatrix} = \begin{pmatrix} 0.0818 & 0.0364 \\ 0.0364 & 0.1273 \end{pmatrix} \begin{pmatrix} P_2 \\ P_3 \end{pmatrix} \tag{9.8}$$

We can write down (9.8) as a general form

$$\delta_{\mathrm{red}} = X_{\mathrm{red}} P_{\mathrm{red}} \tag{9.9}$$

$$X_{\mathrm{red}} = B_{\mathrm{red}}^{-1} \tag{9.10}$$

where B_{red} is the B matrix after eliminating the row and column of slack bus. If the network has n buses then B_{red} would be a square $(n-1) \times (n-1)$ matrix. For the rest of this chapter, whenever X appears in any equation it is representing X_{red}.

Now we are about to investigate the impact of change in power injection in bus n on the line flow between bus i and bus j. The flow on line connecting bus i to bus j is calculated as follows:

$$f_{ij} = \frac{\delta_i - \delta_j}{x_{ij}} \tag{9.11}$$

Now we assume that any change in injected power at bus m will be compensated by the slack bus. In order to calculate the flow change in line $i-j$ we need to use the following equation:

$$\Delta f_{ij} = \frac{\Delta\delta_i - \Delta\delta_j}{x_{ij}} \tag{9.12}$$

The value of x_{ij} remains constant but the voltage angles would change if ΔP is happening at bus m (and $-\Delta P$ at slack bus). Referring to (9.9) we would have

$$\Delta\delta = X \Delta P \tag{9.13}$$

$$\begin{pmatrix} \Delta\delta_2 \\ \Delta\delta_3 \\ \vdots \\ \Delta\delta_m \\ \vdots \\ \Delta\delta_i \\ \vdots \\ \Delta\delta_j \\ \vdots \\ \Delta\delta_{n-1} \\ \Delta\delta_n \end{pmatrix}_{n-1\times 1} = \begin{pmatrix} X_{22} & \cdots & X_{2,n-1} & X_{2n} \\ X_{32} & \cdots & X_{3,n-1} & X_{3n} \\ \vdots & \ddots & \ddots & \vdots \\ X_{m2} & \cdots & X_{m,n-1} & X_{mn} \\ \vdots & \ddots & \ddots & \vdots \\ X_{i2} & \cdots & X_{i,n-1} & X_{in} \\ \vdots & \ddots & \ddots & \vdots \\ X_{j2} & \cdots & X_{j,n-1} & X_{jn} \\ \vdots & \ddots & \ddots & \vdots \\ X_{n-1,2} & \cdots & X_{n-1,n-1} & X_{n-1,n} \\ X_{n,2} & \cdots & X_{n,n-1} & X_{n,n} \end{pmatrix}_{n-1\times n-1} \begin{pmatrix} 0 \\ 0 \\ \vdots \\ \Delta P_m \\ \vdots \\ 0 \\ \vdots \\ 0 \\ \vdots \\ 0 \\ 0 \end{pmatrix}_{n-1\times 1} \tag{9.14}$$

The changes of voltage angles in bus i and j are calculated as follows:

$$\Delta\delta_i = X_{im}\Delta P_m \tag{9.15}$$

$$\Delta\delta_j = X_{jm}\Delta P_m \tag{9.16}$$

$$\Delta f_{ij} = \frac{X_{im}\Delta P_m - X_{jm}\Delta P_m}{x_{ij}} \tag{9.17}$$

The (9.17) states that the sensitivity of the flow in line ij of power change in bus m is obtained as follows:

$$a_m^{ij} = \frac{\Delta f_{ij}}{\Delta P_m} = \frac{X_{im} - X_{jm}}{x_{ij}} \tag{9.18}$$

Using the (9.18) is useful in calculating the line flows in post-contingency period.

$$f_{ij}^{\text{post}} = f_{ij}^{\text{pre}} + a_m^{ij}\Delta P_m \tag{9.19}$$

If the post-contingency line flow at line ij after the failure of a generating unit at bus m (producing P_m^g MW) is to be calculated then the following equation can be used:

$$f_{ij}^{\text{post}} = f_{ij}^{\text{pre}} + a_m^{ij}(-P_m^g) \tag{9.20}$$

If the post-contingency line flow at line ij after the disconnection of load at bus m is to be calculated then the following equation can be used:

$$f_{ij}^{\text{post}} = f_{ij}^{\text{pre}} + a_m^{ij}(L_m) \tag{9.21}$$

Now let's calculate the Generation Shift Factors (a_m^{ij}) for the network shown in Fig. 9.7.

The GCode 9.3 described the GAMS code for calculating the generation shift factors.

GCode 9.3 Generation Shift Factor Calculation

```
Sets bus  /1*3/ ,slack(bus) /1/,Gen /g1*g3/, nonslack(bus) /2*3/ ;
scalars  Sbase /100/ ;
alias(bus,node,shin,knot); alias(nonslack,nonslackj);
Table branch(bus,node,*)    Network technical characteristics
                 X     LIMIT  stat
1    .    2     0.1   100    1
1    .    3     0.2   80     1
2    .    3     0.25  100    1;
set conex(bus,node)          Bus connectivity matrixl;
conex(bus,node)$(branch(bus,node,'x'))=yes;
conex(bus,node)$conex(node,bus)=yes;
branch(bus,node,'x')$branch(node,bus,'x')=branch(node,bus,'x');
branch(bus,node,'stat')$branch(node,bus,'stat')=branch(node,bus,'stat');
branch(bus,node,'Limit')$(branch(bus,node,'Limit')=0)=branch(node,bus,'
     Limit');
branch(bus,node,'bij')$conex(bus,node) =1/branch(bus,node,'x');
Parameter Bmatrix(bus,node),Binv(bus,node),Flow(bus,node); Alias(bus,
     knot);
Bmatrix(bus,node)$(conex(node,bus))=-branch(bus,node,'bij');
Bmatrix(bus,bus)=sum(knot$conex(knot,bus),-Bmatrix(bus,knot));
parameter Breduced(nonslack,nonslackj),GSHF(bus,node,knot);
Breduced(nonslack,nonslackj)=Bmatrix(nonslack,nonslackj);
parameter inva(nonslack,nonslackj) 'inverse of a';

execute_unload 'a.gdx',nonslack,Breduced;
execute '=invert.exe a.gdx nonslack Breduced b.gdx inva';
execute_load 'b.gdx',inva;

Binv(nonslack,nonslackj)=inva(nonslack,nonslackj);
GSHF(bus,node,knot)$conex(bus,node)=
branch(bus,node,'bij')*(Binv(bus,knot)-Binv(node,knot));

Display Bmatrix,Binv,GSHF;
```

The GCode 9.3 has no solve statement or variable. This is because no optimization is going to be done. First of all, the B matrix is calculated in GCode 9.3 as follows:

```
Parameter Bmatrix(bus,node);
Alias(bus,knot);
Bmatrix(bus,node)$conex(node,bus)=-branch(bus,node,'bij');
Bmatrix(bus,bus)=sum(knot$conex(knot,bus),-Bmatrix(bus,knot));
```

This would calculate the Bmatrix as follows:

```
---- 68 PARAMETER Bmatrix
        1          2          3
 1   15.000    -10.000    -5.000
 2  -10.000     14.000    -4.000
 3   -5.000     -4.000     9.000
```

Now we need to eliminate the row and column containing the slack bus. In order to do this, another parameter called Breduced is defined but over the non-slack bus set. This set is defined over all buses except the slack buses.

```
Parameter Breduced(nonslack,nonslackj);
Breduced(nonslack,nonslackj)=Bmatrix(nonslack,nonslackj);
Parameter inva(nonslack,nonslackj) 'inverse of a';
execute_unload 'a.gdx', nonslack, Breduced;
execute '=invert.exe a.gdx nonslack Breduced b.gdx inva';
execute_load 'b.gdx', inva;
```

This would calculate the inverse matrix of the reduced B matrix.

```
---- 68 PARAMETER Binv
        2        3
 2   0.082    0.036
 3   0.036    0.127
```

Now all needed data for calculating the a_m^{ij} is available. The generation shift factors are calculated as follows:

```
Binv(nonslack,nonslackj)=inva(nonslack,nonslackj);
GSHF(bus,node,knot)$conex(bus,node)=branch(bus,node,'bij')*(Binv
(bus,knot)-Binv(node,knot));
```

Using the GCode 9.3, the generation shift factors are calculated as Table 9.2.

The generation shift factors (GSHF) have the following interesting features as follows:

- The procedure for calculating the GSHF does not involve any optimization
- The values of GSHF can be calculated in advance and be used in real-time applications.
- The values of GSHF do not depend on the loading condition of the network. These coefficients only depend on the network topology. If the network topology is changed (due to transmission outage or switching), GSHF should be recalculated.
- The technique we used for calculating the GSHF is assuming that the changes at any bus are quickly compensated by the slack bus. In case the changes are compensated by multiple generating units then the calculation procedure would be slightly different [16].
- The GSHF are also useful for understanding how to reduce the line loading. For example, suppose we need to reduce the flow at line $3-1$. As we can see that bus 3 has the largest GSHF equal to 0.636. It means that if we can reduce the generation at bus 3, then a negative value will be added to the initial flow of line $3-1$. If the initial flow from bus 3 to bus 1 is positive, then it would reduce.

Now let's check the values obtained in Table 9.2. For this purpose, the base power flow is solved and is shown in Fig. 9.8. The values of line flow, as well as the voltage angles, are specified in this figure. Suppose it is desired to increase the flow on the line connecting the bus 3 to bus 2. The power flow on this line is $f_{32}^0 = 2.7\,\text{MW}$. Using the Table 9.2 states that $a_2^{32} = -0.182$ and $a_3^{32} = 0.364$. Let's investigate the impact of changes in power injections in different buses on transmission lines.

9.2.1.1 Demand Increase in L_2 by 10 MW

The demand in bus 2 is increased by 10 MW and the new line flows are depicted in Fig. 9.9. As it can be seen in this figure, the new line flow of line $3-2$ is 4.5 MW. This is obtained using a GAMS code. Let's calculate the new line flow using the Table 9.2. The new line flow is calculated as follows:

$$f_{32}^{\text{post}} = f_{32}^{\text{pre}} + a_2^{32}(\Delta P_2) = 0.027 - 0.182 * (-0.1) = 0.0452 \text{ pu} \tag{9.22}$$

$$f_{12}^{\text{post}} = f_{12}^{\text{pre}} + a_2^{12}(\Delta P_2) = -0.177 - 0.818 * (-0.1) = -0.0952 \text{ pu} \tag{9.23}$$

Table 9.2 Generator shift factors (a_m^{ij}) for three-bus network

Line	Bus (m)	
ij	2	3
1–2	−0.818	−0.364
1–3	−0.182	−0.636
2–1	0.818	0.364
2–3	0.182	−0.364
3–1	0.182	0.636
3–2	−0.182	0.364



Fig. 9.8 The DC power flow solution for three-bus network



Fig. 9.9 The DC power flow solution for three-bus network after increasing the demand at bus 2 for 10 MW.

$$f_{31}^{\text{post}} = f_{31}^{\text{pre}} + a_2^{31}(\Delta P_2) = 0.123 + 0.182 * (-0.1) = 0.1048 \text{ pu} \tag{9.24}$$

It should be noted that the flow values as well as the change in power injection at bus 2 are expressed in pu. ΔP_2 is representing the change in bus injection at bus 2 which is -10 MW or 0.1 pu. It can be observed that the results confirm what is obtained by solving the DC power flow as shown in Fig. 9.9.



Fig. 9.10 The DC power flow solution for three-bus network after increasing the generation at bus 3 for 10 MW.

9.2.1.2 Generation Increase in P_3^g by 10 MW

The generation in bus 3 is increased by 10 MW and the new line flows are depicted in Fig. 9.10 which are obtained using a GAMS code. Let's calculate the new line flow using the GSHF in Table 9.2. The new line flow is calculated as follows:

$$f_{32}^{\text{post}} = f_{32}^{\text{pre}} + a_3^{32}(\Delta P_3) = 0.027 + 0.364 * (0.1) = -0.0634 \text{ pu} \tag{9.25}$$

$$f_{12}^{\text{post}} = f_{12}^{\text{pre}} + a_3^{12}(\Delta P_3) = -0.177 + 0.364 * (0.1) = -0.2134 \text{ pu} \tag{9.26}$$

$$f_{31}^{\text{post}} = f_{31}^{\text{pre}} + a_3^{31}(\Delta P_3) = 0.123 + 0.636 * (0.1) = 0.1866 \text{ pu} \tag{9.27}$$

9.2.2 Line Outage Distribution Factors

The impact of line outages on power flow of other lines is investigated in this section. Consider the line connecting the bus n and m as shown in Fig. 9.11. In Fig. 9.11a, the intact network is shown. We need to find out the impact of the line outage of the branch connecting bus n to bus m on the rest of the network. Suppose the flow of this line is initially equal to f_{nm}^0. The power from the rest of the network injected to bus n is equal to power absorption from bus m to the rest of network when no contingency has happened. We need to find a way to make these flows equal to zero. It is done using a very smart trick [17]. If we add two injections to the network: $+\Delta P_n$ at bus n and another one equal to $-\Delta P_n$ at bus m the flow on the line nm would change as shown in Fig. 9.11b. The question is what is the new flow on this line? The change of flow on line nm can be easily calculated using the following equation:



Fig. 9.11 Line outage modeling using virtual injections. (**a**) Intact network. (**b**) Post-contingency network

$$\Delta f_{nm} = \frac{\Delta\delta_n - \Delta\delta_m}{x_{nm}} = \frac{X_{nn}\Delta P_n + X_{nm}(-\Delta P_n) - (X_{mn}\Delta P_n + X_{mm}(-\Delta P_n))}{x_{nm}} \tag{9.28}$$

This means that

$$\Delta f_{nm} = \frac{X_{nn} + X_{mm} - 2 * X_{nm}}{x_{nm}} \Delta P_n \tag{9.29}$$

Now the post-contingency line flow is calculated as follows:

$$f^{\text{post}}_{nm} = f^0_{nm} + \Delta f_{nm} \tag{9.30}$$

If the virtual injection to the grid ΔP_n is carefully chosen then $f^{\text{post}}_{nm} = \Delta P_n$. This makes the flow from the rest of the network to bus n and m equal to zero (line outage).

$$\Delta P_n = f^{\text{post}}_{nm} = f^0_{nm} + \Delta f_{nm} \tag{9.31}$$

Combining the (9.29) with (9.31) gives us :

$$\Delta P_n = f^0_{nm} + \frac{X_{nn} + X_{mm} - 2X_{nm}}{x_{nm}} \Delta(P_n) \tag{9.32}$$

$$\Delta P_n \left(1 - \frac{X_{nn} + X_{mm} - 2X_{nm}}{x_{nm}}\right) = f^0_{nm}$$

$$\Delta P_n = \frac{f^0_{nm}}{\left(1 - \frac{X_{nn}+X_{mm}-2X_{nm}}{x_{nm}}\right)}$$

Now the change in power flow in line ij is calculated as follows:

$$\Delta f_{ij} = \frac{\Delta\delta_i - \Delta\delta_j}{x_{ij}} = \frac{X_{in}\Delta P_n + X_{im}(-\Delta P_n) - (X_{jn}\Delta P_n + X_{jm}(-\Delta P_n))}{x_{ij}} \tag{9.33}$$

$$\Delta f_{ij} = \frac{X_{in} - X_{im} - X_{jn} + X_{jm}}{x_{ij}} \Delta P_n \tag{9.34}$$

$$\mathrm{LODF}_{ij,nm} = \frac{\Delta f_{ij}}{f^0_{nm}} = \frac{X_{in} - X_{im} - X_{jn} + X_{jm}}{x_{ij}(1 - \frac{X_{nn}+X_{mm}-2X_{nm}}{x_{nm}})} \tag{9.35}$$

The GCode 9.4 is developed to calculate the Line outage distribution factors for the network shown in Fig. 9.7.

GCode 9.4 Line outage distribution factor calculation

```
Sets
bus   /1*3/  ,
slack(bus)  /1/,
Gen  /g1*g3/,
nonslack(bus)  /2*3/  ;

Scalars   Sbase  /100/  ;

Alias(bus,node,shin,knot);
Alias(nonslack,nonslackj);

Table  branch(bus,node,*)      Network  technical  characteristics
                X      LIMIT  stat
1      .      2      0.1    100     1
1      .      3      0.2    80      1
2      .      3      0.25   100     1;

Set  conex(bus,node)           Bus  connectivity  matrixl;

conex(bus,node)$(branch(bus,node,'x'))=yes;
conex(bus,node)$conex(node,bus)=yes;

branch(bus,node,'x')$branch(node,bus,'x')=branch(node,bus,'x');
branch(bus,node,'stat')$branch(node,bus,'stat')=branch(node,bus,'stat');
branch(bus,node,'Limit')$(branch(bus,node,'Limit')=0)=branch(node,bus,'
      Limit');
branch(bus,node,'bij')$conex(bus,node)  =1/branch(bus,node,'x');

Parameter  Bmatrix(bus,node),Binv(bus,node);
Alias(bus,knot);

Bmatrix(bus,node)$(conex(node,bus))=-branch(bus,node,'bij');
Bmatrix(bus,bus)=sum(knot$conex(knot,bus),-Bmatrix(bus,knot));
parameter  Breduced(nonslack,nonslackj),GSHF(bus,node,knot),X0(bus,node);
Breduced(nonslack,nonslackj)=Bmatrix(nonslack,nonslackj);

Parameter  inva(nonslack,nonslackj)  'inverse  of  a',Dfactor(bus,node,knot,
      shin),
```

```
contingency(bus,node,knot,shin);

execute_unload 'a.gdx',nonslack,Breduced;
execute '=invert.exe a.gdx nonslack Breduced b.gdx inva';
execute_load 'b.gdx',inva;

Binv(nonslack,nonslackj)=inva(nonslack,nonslackj);
GSHF(bus,node,knot)$conex(bus,node)=
branch(bus,node,'bij')*(Binv(bus,knot)-Binv(node,knot));

dfactor(bus,node,knot,shin)$( conex(bus,node) and  conex(knot,shin)
and (ord(bus)>ord(node))
and (ord(knot)>ord(shin))
and (ord(bus)<>ord(knot) or  ord(node)<>ord(shin)))=
branch(bus,node,'bij')*(Binv(bus,knot)-Binv(bus,shin)-Binv(node,knot)
+Binv(node,shin))/(1-branch(knot,shin,'bij')*(Binv(knot,knot)
+Binv(shin,shin)-2*Binv(knot,shin)));

Display Bmatrix,Binv,GSHF,dfactor;
```

Line outage distribution factors ($\mathrm{LODF}_{ij,nm}$) are described in Table 9.3.

The branch data for IEEE RTS 24-bus network is provided in Table 9.4. The LODF and GSF coefficients are calculated for IEEE RTS 24-bus (Fig. 9.12) and are given in Tables 9.5 and 9.6, respectively.

9.3 Transmission Network Switching

The idea of optimal transmission switching (OTS) has been broadly investigated in the literature [18]. Opening a set of transmission lines would change the network topology and the line flow patterns. This can be used to relieve the line congestion in the system and reduce the operating costs. Some research works which used the transmission switching as a flexibility tool are listed as follows:

Table 9.3 Line outage distribution factors ($\mathrm{LODF}_{ij,nm}$) for three-bus network

Line	*m*		
ij	*n*	1	2
2–1	3	1	−1
3–1	2	1	
3–1	3		1
3–2	2	−1	
3–2	3	1	

Table 9.4 Branch data for IEEE RTS 24-bus network

From	To	r(pu)	x(pu)	b(pu)	Rating (MVA)	From	To	r(pu)	x(pu)	b(pu)	Rating (MVA)
1	2	0.0026	0.0139	0.4611	175	11	13	0.0061	0.0476	0.0999	500
1	3	0.0546	0.2112	0.0572	175	11	14	0.0054	0.0418	0.0879	500
1	5	0.0218	0.0845	0.0229	175	12	13	0.0061	0.0476	0.0999	500
2	4	0.0328	0.1267	0.0343	175	12	23	0.0124	0.0966	0.2030	500
2	6	0.0497	0.1920	0.0520	175	13	23	0.0111	0.0865	0.1818	500
3	9	0.0308	0.1190	0.0322	175	14	16	0.0050	0.0389	0.0818	500
3	24	0.0023	0.0839	0.0000	400	15	16	0.0022	0.0173	0.0364	500
4	9	0.0268	0.1037	0.0281	175	15	21	0.0032	0.0245	0.2060	1000
5	10	0.0228	0.0883	0.0239	175	15	24	0.0067	0.0519	0.1091	500
6	10	0.0139	0.0605	2.4590	175	16	17	0.0033	0.0259	0.0545	500
7	8	0.0159	0.0614	0.0166	175	16	19	0.0030	0.0231	0.0485	500
8	9	0.0427	0.1651	0.0447	175	17	18	0.0018	0.0144	0.0303	500
8	10	0.0427	0.1651	0.0447	175	17	22	0.0135	0.1053	0.2212	500
9	11	0.0023	0.0839	0.0000	400	18	21	0.0017	0.0130	0.1090	1000
9	12	0.0023	0.0839	0.0000	400	19	20	0.0026	0.0198	0.1666	1000
10	11	0.0023	0.0839	0.0000	400	20	23	0.0014	0.0108	0.0910	1000
10	12	0.0023	0.0839	0.0000	400	21	22	0.0087	0.0678	0.1424	500



Fig. 9.12 IEEE RTS 24-bus network

Table 9.5 $\mathrm{LODF}_{12,\ell}$ calculated for IEEE RTS 24-bus network

$\ell = nm$	1	2	3	4	5	6	8	9	10	11	12	13	14	15	16	17	18	19	20	21
3	0.611																			
4		−0.712																		
5	0.706																			
6		−0.712																		
9			0.291	−0.712			0.054													
10					0.706	−0.712	−0.054													
11								−0.067	0.022											
12								−0.062	0.028											
13										−0.002	−0.010									
14										−0.048										
16													−0.048	0.031						
17															−0.009					
18																−0.006				
19															0.043					
20																		0.043		
21														0.009			−0.006			
22																−0.003				0.003
23											− 0.023	− 0.016							0.043	
24			0.154											−0.154						

Table 9.6 $GSHF_{\ell,m}$ calculated for IEEE RTS 24-bus network

Line	Bus (m)																					
ℓ_{ij}	1	2	3	4	5	6	7	8	9	10	11	12	14	15	16	17	18	19	20	21	22	23	24
2 1	−0.44	0.51	−0.10	0.25	−0.23	0.11	0.01	0.01	0.04	−0.01	0.00	0.00	−0.01	−0.03	−0.02	−0.02	−0.02	−0.02	−0.01	−0.03	−0.02	−0.01	−0.05
3 1	−0.24	-0.22	0.20	−0.09	−0.16	−0.10	−0.03	−0.03	0.01	−0.06	0.00	−0.01	0.02	0.06	0.04	0.05	0.05	0.03	0.02	0.05	0.05	0.02	0.11
4 2	−0.24	−0.27	−0.03	0.37	−0.15	−0.11	0.00	0.00	0.07	−0.06	0.00	0.00	0.00	−0.01	−0.01	−0.01	−0.01	0.00	0.00	−0.01	−0.01	0.00	−0.02
5 1	−0.32	−0.29	−0.10	−0.15	0.38	−0.01	0.01	0.01	−0.05	0.07	0.00	0.00	−0.01	−0.03	−0.02	−0.03	−0.03	−0.02	−0.01	−0.03	−0.03	−0.01	−0.06
6 2	−0.20	−0.22	−0.07	−0.12	−0.07	0.22	0.01	0.01	−0.03	0.05	0.00	0.00	−0.01	−0.02	−0.01	−0.02	−0.02	−0.01	−0.01	−0.02	−0.02	−0.01	−0.04
8 7							−1.00																
9 3	−0.09	−0.07	−0.43	0.03	−0.05	−0.02	0.06	0.06	0.12	−0.01	0.00	0.02	−0.05	−0.12	−0.09	−0.10	−0.11	−0.07	−0.05	−0.11	−0.11	−0.04	−0.24
9 4	−0.24	−0.27	−0.03	−0.63	−0.15	−0.11	0.00	0.00	0.07	−0.06	0.00	0.00	0.00	−0.01	−0.01	−0.01	−0.01	0.00	0.00	−0.01	−0.01	0.00	−0.02
9 8	−0.02	−0.02	0.03	0.03	−0.05	−0.07	−0.50	−0.50	0.07	−0.08	0.00	0.00	0.00	0.01	0.01	0.01	0.01	0.00	0.00	0.01	0.01	0.00	0.02
10 5	−0.32	−0.29	−0.10	−0.15	−0.62	−0.01	0.01	0.01	−0.05	0.07	0.00	0.00	−0.01	−0.03	−0.02	−0.03	−0.03	−0.02	−0.01	−0.03	−0.03	−0.01	−0.06
10 6	−0.20	−0.22	−0.07	−0.12	−0.07	−0.78	0.01	0.01	−0.03	0.05	0.00	0.00	−0.01	−0.02	−0.01	−0.02	−0.02	−0.01	−0.01	−0.02	−0.02	− 0.01	−0.04
10 8	0.02	0.02	−0.03	−0.03	0.05	0.07	−0.50	−0.50	−0.07	0.08	0.00	0.00	0.00	−0.01	−0.01	−0.01	−0.01	0.00	0.00	−0.01	−0.01	0.00	−0.02
11 9	−0.17	−0.17	−0.20	−0.28	−0.12	−0.09	−0.22	−0.22	−0.36	−0.07	0.12	−0.10	0.05	−0.03	−0.01	−0.02	−0.02	−0.02	−0.03	−0.02	−0.02	−0.03	−0.09
11 10	−0.24	−0.24	−0.08	−0.15	−0.31	−0.35	−0.23	−0.23	−0.07	−0.39	0.11	−0.10	0.06	0.00	0.01	0.01	0.01	0.00	−0.01	0.01	0.01	−0.02	−0.03
12 9	−0.18	−0.19	−0.23	−0.29	−0.13	−0.11	−0.23	−0.23	−0.38	−0.08	−0.11	0.12	−0.10	−0.10	−0.08	−0.09	−0.09	−0.05	−0.02	−0.09	−0.09	−0.01	−0.15
12 10	−0.26	−0.25	−0.12	−0.16	−0.33	−0.37	−0.24	−0.24	−0.08	−0.40	−0.11	0.11	−0.08	−0.06	−0.06	−0.06	−0.06	−0.03	−0.01	−0.06	−0.06	0.00	−0.08
13 11	−0.43	−0.43	−0.42	−0.43	−0.43	−0.43	−0.43	−0.43	−0.43	−0.43	−0.63	−0.23	−0.51	−0.41	−0.40	−0.40	−0.40	−0.34	−0.28	−0.40	−0.40	−0.24	−0.41
13 12	−0.40	−0.40	−0.36	−0.40	−0.41	−0.41	−0.41	−0.41	−0.41	−0.41	−0.23	−0.61	−0.26	−0.29	−0.28	−0.28	−0.29	−0.28	−0.28	−0.29	−0.29	−0.29	−0.32
14 11	0.02	0.02	0.14	0.01	0.00	−0.02	−0.02	−0.02	0.00	−0.03	−0.14	0.03	0.63	0.38	0.40	0.40	0.39	0.31	0.24	0.39	0.39	0.19	0.29
16 14	0.02	0.02	0.14	0.01	0.00	−0.02	−0.02	−0.02	0.00	−0.03	−0.14	0.03	−0.37	0.38	0.40	0.40	0.39	0.31	0.24	0.39	0.39	0.19	0.29
16 15	−0.12	−0.12	−0.30	−0.10	−0.09	−0.06	−0.07	−0.07	−0.09	−0.05	0.00	−0.02	0.06	−0.66	0.11	−0.16	−0.29	0.08	0.06	−0.41	−0.31	0.05	−0.52

(continued)

Table 9.6 (continued)

Line	Bus (m)																					
ℓ_{ij}	1	2	3	4	5	6	7	8	9	10	11	12	14	15	16	17	18	19	20	21	22	23	24
21 15	−0.03	−0.03	−0.07	−0.02	−0.02	−0.02	−0.02	−0.02	−0.02	−0.01	0.00	0.00	0.01	−0.15	0.03	0.31	0.45	0.02	0.01	0.57	0.47	0.01	−0.12
21 18	0.02	0.02	0.06	0.02	0.02	0.01	0.01	0.01	0.02	0.01	0.00	0.00	−0.01	0.13	−0.02	−0.27	−0.46	−0.02	−0.01	0.37	0.12	−0.01	0.11
22 17	0.00	0.00	0.01	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.02	0.00	−0.04	0.01	0.00	0.00	0.06	0.41	0.00	0.02
22 21	0.00	0.00	−0.01	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	−0.02	0.00	0.04	−0.01	0.00	0.00	−0.06	0.59	0.00	−0.02
23 12	−0.04	−0.04	0.02	−0.05	−0.05	−0.06	−0.06	−0.06	−0.05	−0.07	0.01	−0.16	0.08	0.13	0.14	0.14	0.14	0.20	0.25	0.14	0.14	0.28	0.09
23 13	0.17	0.17	0.22	0.17	0.16	0.16	0.16	0.16	0.16	0.15	0.13	0.16	0.23	0.31	0.31	0.31	0.31	0.38	0.44	0.31	0.31	0.47	0.27
23 20	−0.13	−0.13	−0.23	−0.12	−0.11	−0.10	−0.10	−0.10	−0.11	−0.09	−0.14	0.00	−0.31	−0.44	−0.46	−0.45	−0.45	−0.58	−0.69	−0.44	−0.45	0.25	−0.36
24 3	−0.15	−0.15	−0.37	−0.13	−0.11	−0.08	−0.08	−0.08	−0.11	−0.06	−0.01	−0.03	0.07	0.18	0.14	0.15	0.16	0.10	0.07	0.17	0.16	0.06	0.35
24 15	0.15	0.15	0.37	0.13	0.11	0.08	0.08	0.08	0.11	0.06	0.01	0.03	−0.07	−0.18	−0.14	−0.15	−0.16	−0.10	−0.07	−0.17	−0.16	−0.06	0.65

- DC-OPF considering $N-1$ contingencies [19]
- Co-optimization of unit commitment and transmission switching with $N-1$ reliability constraints [20]
- Robust transmission switching considering $N-k$ contingencies [21]
- Optimal transmission switching considering short-circuit current limitation constraints [22]
- Probabilistic security analysis of OTS [23]
- Chance-constrained OTS with guaranteed wind power utilization [24]
- Heuristics OTS based on DC-OPF and AC-OPF [25]

In this section, the OTS is solved using GAMS and then we will discuss how this model can be improved and extended. The transmission switching problem is formulated as a MIP model in (9.36).

$$\text{OF} = \sum_{g \in \Omega_G} b_g P_g \tag{9.36a}$$

$$P_{ij} - B_{ij}(\delta_i - \delta_j) \leq (1 - \zeta_{ij})M \tag{9.36b}$$

$$P_{ij} - B_{ij}(\delta_i - \delta_j) \geq -(1 - \zeta_{ij})M \tag{9.36c}$$

$$\sum_{g \in \Omega_G^i} P_g + \text{LS}_i - L_i = \sum_{j \in \Omega_\ell^i} P_{ij} : \lambda_i \quad i \in \Omega_B \tag{9.36d}$$

$$-P_{ij}^{\max}\zeta_{ij} \leq P_{ij} \leq P_{ij}^{\max}\zeta_{ij} \quad ij \in \Omega_\ell \tag{9.36e}$$

$$P_g^{\min} \leq P_g \leq P_g^{\max} \tag{9.36f}$$

$$\sum_{ij}(1 - \zeta_{ij}) \leq N_{\text{sw}} \quad ij \in \Omega_\ell \tag{9.36g}$$

$$B_{ij} = \frac{1}{x_{ij}} \tag{9.36h}$$

$$\zeta_{ij} \in \{0, 1\} \tag{9.36i}$$

where ζ_{ij} is the on/off state of the branch connecting bus i to bus j, and N_{sw} is the number of allowed switching actions in the network.

The operating and congestion cost vs number of switched lines in IEEE 118-bus network (Fig. 9.13) are depicted in Fig. 9.14. The GAMS code for solving the (9.36) is provided in GCode 9.5.



Fig. 9.13 IEEE 118-bus network



Fig. 9.14 The operating and congestion cost vs number of switched lines in IEEE 118-bus network

GCode 9.5 The OTS GAMS code for IEEE 118-bus network

```
Sets bus /1*118/, slack(bus) /13/, conex(bus,node),
GenNo /Gen1*Gen54/,counter /c0*c10/;
Scalars    Sbase /100/;
Alias(bus,totalbus,node);

Table GenDatanew(bus,GenNo,*)
                b          pmin pmax ;
Table BusData(bus,*) buss characteristics
          Pd ;
Table branch(bus,totalbus,*)
              x          Ilim ;
branch(bus,totalbus,'bij')$branch(totalbus,bus,'x') =1/branch(bus,
    totalbus,'x');
conex(bus,node)$branch(bus,node,'x')=yes;
parameter branch(bus,totalbus,*),M,NSW,report(counter,*);
M=smax((bus,node)$conex(bus,node), branch(bus,node,'bij')*2*pi/3);
Positive variable Pg(GenNo);
Variables Pij(bus,node),delta(bus),ROF;
BINARY VARIABLE SW(bus,node);
Equations const0,const1,const2,const3,const0A,
const0B,const0C,const0D,const0E,const0F;
const0(bus,node)$conex(bus,node) .. Pij(bus,node)=e=
branch(bus,node,'bij')*(delta(bus)-delta(node));
const1(bus).. sum(GenNo$GenDatanew(bus,GenNo,'Pmax'),Pg(GenNo))
```

```
-BusData(bus,'Pd')/sbase=e=+sum(node$conex(node,bus),Pij(bus,node));
const2       .. ROF=g= sum((GenNo,bus)$GenDatanew(bus,GenNo,'Pmax'),
GenDatanew(bus,GenNo,'b')*Pg(GenNo)*Sbase);
const0A(bus,node)$conex(bus,node)..
Pij(bus,node)-branch(bus,node,'bij')*(delta(bus)-delta(node))=l= M*(1-SW
     (bus,node));
const0B(bus,node)$conex(bus,node)..
Pij(bus,node)-branch(bus,node,'bij')*(delta(bus)-delta(node))=g=-M*(1-SW
     (bus,node));
const0C(bus,node)$conex(bus,node)..
Pij(bus,node)=l= SW(bus,node)*branch(bus,node,'Ilim');
const0D(bus,node)$conex(bus,node)..
Pij(bus,node)=g=-SW(bus,node)*branch(bus,node,'Ilim');
const0E(bus,node)$conex(bus,node)..SW(node,bus)=e=SW(bus,node);
const0F(bus,node)$conex(bus,node).. Pij(node,bus)=e=-Pij(bus,node);
const3 ..    0.5*sum((bus,node)$conex(bus,node),1-SW(bus,node))=l=NSW;
model BASE /const0,const1,const2/;
model Switching/const1,const2,const0A,const0B,const0C,
                                        const0D,const0E,const0F,const3
                                            /;
Option  Optca=0; Option  Optcr=0;

BusData(bus,'Pd')=1.1*BusData(bus,'Pd');
Pg.lo(GenNo)=sum(bus,GenDatanew(bus,GenNo,'Pmin'))/Sbase;
Pg.up(GenNo)=sum(bus,GenDatanew(bus,GenNo,'Pmax'))/Sbase;
delta.up(bus)=pi/3;  delta.lo(bus)=-pi/3;  delta.l(bus)=0;  delta.fx(slack)
     =0;
Pij.up(bus,node)$conex(bus,node)= 1*branch(bus,node,'Ilim');
Pij.lo(bus,node)$conex(bus,node)=-1*branch(bus,node,'Ilim');
Solve BASE minimizing ROF using lp;
SW.l(bus,node)=1;  report('c0','OF')=ROF.l;
report('c0','NSW')=0.5*sum((bus,node)$conex(bus,node),1-SW.l(bus,node));
report('c0','Congestion')=0.5*sum((bus,node)$conex(bus,node),
(-const1.m(bus)+const1.m(node))*Pij.l(bus,node));
loop(counter $(ord(counter)>1),
NSW=ord(counter)-1;
Solve switching minimizing ROF using mip;
report(counter,'OF')=ROF.l;
report(counter,'NSW')=0.5*sum((bus,node)$conex(bus,node),1-SW.l(bus,node
     ));
report(counter,'Congestion')=0.5*sum((bus,node)$conex(bus,node),
                              (-const1.m(bus)+const1.m(node))*Pij.l(bus,node
                                   ));
);
```

Increasing the demand in a given area or bus (or equivalently losing the generation) might cause congestion and increasing the total operating costs. The OTS can be used to enhance the grid utilization and reduce the line congestions. A simple analysis is conducted as follows:

- The demand at bus i is increased for 20 MW.
- The DC OPF is solved without considering the transmission switching option (LP model).
- The DC OPF along with transmission switching option is solved (MIP model).



Fig. 9.15 The operating cost vs the connection point of new demand (20 MW) in with/without transmission switching cases



Fig. 9.16 The LMP values (λ_i) in ($/MW h) at different buses in with/without transmission switching cases

The comparison of the total operating costs between the with/without transmission switching cases are shown in Fig. 9.15. The impact of connecting a new demand to different buses would cause different changes in total operating costs. In all cases, using the transmission switching flexibility can reduce the operating costs as shown in Fig. 9.15.

Using the transmission switching might change the LMP values at different buses. The LMP values (λ_i) in $/MW h at different buses in with/without transmission switching cases. As it is shown in Fig. 9.16, when no switching is allowed ($N_{\text{sw}} = 0$) then the LMP values of switching and not switching cases are the same. By increasing the number of switchable lines, the LMP values get closer to each other. The generation and branch data of IEEE 118 bus are given in Tables 9.7 and 9.8, respectively.

The developed GAMS code for OTS can be improved to consider the following issues:

Table 9.7 Generation data for 118-bus network

Bus	Unit (g)	b_g (\$/MW h)	$P_g^{\min}$ (MW)	$P_g^{\max}$ (MW)	Bus	Unit (g)	b_g (\$/MW h)	$P_g^{\min}$ (MW)	$P_g^{\max}$ (MW)
1	Gen1	26.2	5.0	30.0	65.0	Gen28	8.3	100.0	420.0
4	Gen2	26.2	5.0	30.0	66.0	Gen29	12.9	80.0	300.0
6	Gen3	26.2	5.0	30.0	69.0	Gen30	15.5	30.0	80.0
8	Gen4	12.9	150.0	300.0	70.0	Gen31	26.2	10.0	30.0
10	Gen5	12.9	100.0	300.0	72.0	Gen32	26.2	5.0	30.0
12	Gen6	26.2	10.0	30.0	73.0	Gen33	37.7	5.0	20.0
15	Gen7	17.8	25.0	100.0	74.0	Gen34	17.8	25.0	100.0
18	Gen8	26.2	5.0	30.0	76.0	Gen35	17.8	25.0	100.0
19	Gen9	26.2	5.0	30.0	77.0	Gen36	12.9	150.0	300.0
24	Gen10	12.9	100.0	300.0	80.0	Gen37	17.8	25.0	100.0
25	Gen11	10.8	100.0	350.0	85.0	Gen38	26.2	10.0	30.0
26	Gen12	26.2	8.0	30.0	87.0	Gen39	10.8	100.0	300.0
27	Gen13	26.2	8.0	30.0	89.0	Gen40	12.9	50.0	200.0
31	Gen14	17.8	25.0	100.0	90.0	Gen41	37.7	8.0	20.0
32	Gen15	26.2	8.0	30.0	91.0	Gen42	22.9	20.0	50.0
34	Gen16	17.8	25.0	100.0	92.0	Gen43	12.9	100.0	300.0
36	Gen17	26.2	8.0	30.0	99.0	Gen44	12.9	100.0	300.0
40	Gen18	26.2	8.0	30.0	100.0	Gen45	12.9	100.0	300.0

(continued)

Table 9.7 (continued)

Bus	Unit (g)	b_g (\$/MW h)	$P_g^{\min}$ (MW)	$P_g^{\max}$ (MW)	Bus	Unit (g)	b_g (\$/MW h)	$P_g^{\min}$ (MW)	$P_g^{\max}$ (MW)
42	Gen19	17.8	25.0	100.0	103.0	Gen46	37.7	8.0	20.0
46	Gen20	12.3	50.0	250.0	104.0	Gen47	17.8	25.0	100.0
49	Gen21	12.3	50.0	250.0	105.0	Gen48	17.8	25.0	100.0
54	Gen22	17.8	25.0	100.0	107.0	Gen49	37.7	8.0	20.0
55	Gen23	17.8	25.0	100.0	110.0	Gen50	22.9	25.0	50.0
56	Gen24	13.3	50.0	200.0	111.0	Gen51	17.8	25.0	100.0
59	Gen25	13.3	50.0	200.0	112.0	Gen52	17.8	25.0	100.0
61	Gen26	17.8	25.0	100.0	113.0	Gen53	17.8	25.0	100.0
62	Gen27	8.3	100.0	420.0	116.0	Gen54	22.9	25.0	50.0

Table 9.8 Branch data for 118-bus network

Line	x	Limit	Line	x	Limit	Line	x	Limit	Line	x	Limit	Line	x	Limit	Line	x	Limit
1 . 2	0.10	1.75	23 . 25	0.08	5.00	44 . 45	0.09	1.75	63 . 64	0.02	5.00	68 . 81	0.02	5.00	98 . 100	0.18	1.75
1 . 3	0.04	1.75	26 . 25	0.04	5.00	45 . 46	0.14	1.75	64 . 61	0.03	5.00	81 . 80	0.04	5.00	99 . 100	0.08	1.75
4 . 5	0.01	5.00	25 . 27	0.16	5.00	46 . 47	0.13	1.75	38 . 65	0.10	5.00	77 . 82	0.09	2.00	100 . 101	0.13	1.75
3 . 5	0.11	1.75	27 . 28	0.09	1.75	46 . 48	0.19	1.75	64 . 65	0.03	5.00	82 . 83	0.04	2.00	92 . 102	0.06	1.75
5 . 6	0.05	1.75	28 . 29	0.09	1.75	47 . 49	0.06	1.75	49 . 66	0.05	10.00	83 . 84	0.13	1.75	101 . 102	0.11	1.75
6 . 7	0.02	1.75	30 . 17	0.04	5.00	42 . 49	0.16	3.50	62 . 66	0.22	1.75	83 . 85	0.15	1.75	100 . 103	0.05	5.00
8 . 9	0.03	5.00	8 . 30	0.05	1.75	45 . 49	0.19	1.75	62 . 67	0.12	1.75	84 . 85	0.06	1.75	100 . 104	0.20	1.75
8 . 5	0.03	5.00	26 . 30	0.09	5.00	48 . 49	0.05	1.75	65 . 66	0.04	5.00	85 . 86	0.12	5.00	103 . 104	0.16	1.75
9 . 10	0.03	5.00	17 . 31	0.16	1.75	49 . 50	0.08	1.75	66 . 67	0.10	1.75	86 . 87	0.21	5.00	103 . 105	0.16	1.75
4 . 11	0.07	1.75	29 . 31	0.03	1.75	49 . 51	0.14	1.75	65 . 68	0.02	5.00	85 . 88	0.10	1.75	100 . 106	0.23	1.75
5 . 11	0.07	1.75	23 . 32	0.12	1.40	51 . 52	0.06	1.75	47 . 69	0.28	1.75	85 . 89	0.17	1.75	104 . 105	0.04	1.75
11 . 12	0.02	1.75	31 . 32	0.10	1.75	52 . 53	0.16	1.75	49 . 69	0.32	1.75	88 . 89	0.07	5.00	105 . 106	0.05	1.75
2 . 12	0.06	1.75	27 . 32	0.08	1.75	53 . 54	0.12	1.75	68 . 69	0.04	5.00	89 . 90	0.09	10.00	105 . 107	0.18	1.75
3 . 12	0.16	1.75	15 . 33	0.12	1.75	49 . 54	0.15	3.50	69 . 70	0.13	5.00	90 . 91	0.08	1.75	105 . 108	0.07	1.75
7 . 12	0.03	1.75	19 . 34	0.25	1.75	54 . 55	0.07	1.75	24 . 70	0.41	1.75	89 . 92	0.08	10.00	106 . 107	0.18	1.75
11 . 13	0.07	1.75	35 . 36	0.01	1.75	54 . 56	0.01	1.75	70 . 71	0.04	1.75	91 . 92	0.13	1.75	108 . 109	0.03	1.75
12 . 14	0.07	1.75	35 . 37	0.05	1.75	55 . 56	0.02	1.75	24 . 72	0.20	1.75	92 . 93	0.08	1.75	103 . 110	0.18	1.75
13 . 15	0.24	1.75	33 . 37	0.14	1.75	56 . 57	0.10	1.75	71 . 72	0.18	1.75	92 . 94	0.16	1.75	109 . 110	0.08	1.75
14 . 15	0.20	1.75	34 . 36	0.03	1.75	50 . 57	0.13	1.75	71 . 73	0.05	1.75	93 . 94	0.07	1.75	110 . 111	0.08	1.75
12 . 16	0.08	1.75	34 . 37	0.01	5.00	56 . 58	0.10	1.75	70 . 74	0.13	1.75	94 . 95	0.04	1.75	110 . 112	0.06	1.75
15 . 17	0.04	5.00	38 . 37	0.04	5.00	51 . 58	0.07	1.75	70 . 75	0.14	1.75	80 . 96	0.18	1.75	17 . 113	0.03	1.75
16 . 17	0.18	1.75	37 . 39	0.11	1.75	54 . 59	0.23	1.75	69 . 75	0.12	5.00	82 . 96	0.05	1.75	32 . 113	0.20	5.00

(continued)

Table 9.8 (continued)

Line	x	Limit	Line	x	Limit	Line	x	Limit	Line	x	Limit	Line	x	Limit	Line	x	Limit
17 . 18	0.05	1.75	37 . 40	0.17	1.75	56 . 59	0.13	3.50	74 . 75	0.04	1.75	94 . 96	0.09	1.75	32 . 114	0.06	1.75
18 . 19	0.05	1.75	30 . 38	0.05	1.75	55 . 59	0.22	1.75	76 . 77	0.15	1.75	80 . 97	0.09	1.75	27 . 115	0.07	1.75
19 . 20	0.12	1.75	39 . 40	0.06	1.75	59 . 60	0.15	1.75	69 . 77	0.10	1.75	80 . 98	0.11	1.75	114 . 115	0.01	1.75
15 . 19	0.04	1.75	40 . 41	0.05	1.75	59 . 61	0.15	1.75	75 . 77	0.20	1.75	80 . 99	0.21	2.00	68 . 116	0.00	5.00
20 . 21	0.08	1.75	40 . 42	0.18	1.75	60 . 61	0.01	5.00	77 . 78	0.01	1.75	92 . 100	0.30	1.75	12 . 117	0.14	1.75
21 . 22	0.10	1.75	41 . 42	0.14	1.75	60 . 62	0.06	1.75	78 . 79	0.02	1.75	94 . 100	0.06	1.75	75 . 118	0.05	1.75
22 . 23	0.16	1.75	43 . 44	0.25	1.75	61 . 62	0.04	1.75	77 . 80	0.05	10.00	95 . 96	0.05	1.75	76 . 118	0.05	1.75
23 . 24	0.05	1.75	34 . 43	0.17	1.75	63 . 59	0.04	5.00	79 . 80	0.07	1.75	96 . 97	0.09	1.75	12 . 117	0.14	1.75
															75 . 118	0.05	1.75
															76 . 118	0.05	1.75

- The model should be multi-period. The OTS should consider the variation pattern of demand and then determine the optimal switching actions.
- The uncertainty of demand and renewable power generation should be taken into account.
- The current formulation does not ensure the network connectivity. It only tries to satisfy the nodal demand-supply constraint. The resultant system (after switching) might contain some islands.
- The computation burden of the model should be improved to make it applicable for large scale transmission networks.
- The AC power flow constraints should be used for getting closer to reality.
- The unit commitment constraints can be added to the formulation to consider the on/off states of the units as the decision variables.
- Changing the transmission network topology might change the short circuit level on each bus. This should be taken into account for protection issues.
- The current model only considers the intact condition of the network. The contingencies should also be considered.
- The OTS flexibility can be combined with demand response and power flow controller devices.
- The bus splitting can be regarded as a switching action.

References

1. G.A. Orfanos, P.S. Georgilakis, N.D. Hatziargyriou, Transmission expansion planning of systems with increasing wind power integration. IEEE Trans. Power Syst. **28**(2), 1355–1362 (2013)
2. H. Zhang, V. Vittal, G.T. Heydt, J. Quintero, A mixed-integer linear programming approach for multi-stage security-constrained transmission expansion planning. IEEE Trans. Power Syst. **27**(2), 1125–1133 (2012)
3. I. De J. Silva, M.J. Rider, R. Romero, C.A.F. Murari, Transmission network expansion planning considering uncertainty in demand. IEEE Trans. Power Syst. **21**(4), 1565–1573 (2006)
4. J.D. Finney, H.A. Othman, W.L. Rutz, Evaluating transmission congestion constraints in system planning. IEEE Trans. Power Syst. **12**(3), 1143–1150 (1997)
5. R.A. Jabr, Robust transmission network expansion planning with uncertain renewable generation and loads. IEEE Trans. Power Syst. **28**(4), 4558–4567 (2013)
6. S. Haffner, A. Monticelli, A. Garcia, J. Mantovani, R. Romero, Branch and bound algorithm for transmission system expansion planning using a transportation model. IEE Proc. Gener. Transm. Distrib. **147**(3), 149–156 (2000)
7. A.K. Kazerooni, J. Mutale, Transmission network planning under security and environmental constraints. IEEE Trans. Power Syst. **25**(2), 1169–1178 (2010)
8. P. Maghouli, S.H. Hosseini, M. Oloomi Buygi, M. Shahidehpour, A scenario-based multi-objective model for multi-stage transmission expansion planning. IEEE Trans. Power Syst. **26**(1), 470–478 (2011)
9. J.H. Roh, M. Shahidehpour, L. Wu, Market-based generation and transmission planning with uncertainties. IEEE Trans. Power Syst. **24**(3), 1587–1598 (2009)
10. M.J. Rider, A.V. Garcia, R. Romero, Power system transmission network expansion planning using AC model. IET Gener. Transm. Distrib. **1**(5), 731–742 (2007)

11. P. Maghouli, S.H. Hosseini, M.O. Buygi, M. Shahidehpour, A multi-objective framework for transmission expansion planning in deregulated environments. IEEE Trans. Power Syst. **24**(2), 1051–1061 (2009)
12. H. Yu, C.Y. Chung, K.P. Wong, J.H. Zhang, A chance constrained transmission network expansion planning method with consideration of load and wind farm uncertainties. IEEE Trans. Power Syst. **24**(3), 1568–1576 (2009)
13. M.R. Bussieck, A. Pruessner, Mixed-integer nonlinear programming. SIAG/OPT Newsl. Views News **14**(1), 19–22 (2003)
14. R. Romero, A. Monticelli, A. Garcia, S. Haffner, Test systems and mathematical models for transmission network expansion planning. IEE Proc. Gener. Transm. Distrib. **149**(1), 27–36 (2002)
15. L. Baringo, A.J. Conejo, Transmission and wind power investment. IEEE Trans. Power Syst. **27**(2), 885–893 (2012)
16. A.J. Wood, B.F. Wollenberg, *Power Generation Operation and Control*, 2nd edn. (Wiley, New York, 1996)
17. A.J. Wood, B.F. Wollenberg, *Power Generation, Operation, and Control* (Wiley, New York, 2012)
18. E.B. Fisher, R.P. O'Neill, M.C. Ferris, Optimal transmission switching. IEEE Trans. Power Syst. **23**(3), 1346–1355 (2008)
19. K.W. Hedman, R.P. O'Neill, E.B. Fisher, S.S. Oren, Optimal transmission switching with contingency analysis. IEEE Trans. Power Syst. **24**(3), 1577–1586 (2009)
20. K.W. Hedman, M.C. Ferris, R.P. O'Neill, E.B. Fisher, S.S. Oren, Co-optimization of generation unit commitment and transmission switching with n-1 reliability. IEEE Trans. Power Syst. **25**(2), 1052–1063 (2010)
21. T. Ding, C. Zhao, Robust optimal transmission switching with the consideration of corrective actions for n-k contingencies. IET Gener. Transm. Distrib. **10**(13), 3288–3295 (2016)
22. Z. Yang, H. Zhong, Q. Xia, C. Kang, Optimal transmission switching with short-circuit current limitation constraints. IEEE Trans. Power Syst. **31**(2), 1278–1288 (2016)
23. P. Henneaux, D.S. Kirschen, Probabilistic security analysis of optimal transmission switching. IEEE Trans. Power Syst. **31**(1), 508–517 (2016)
24. F. Qiu, J. Wang, Chance-constrained transmission switching with guaranteed wind power utilization. IEEE Trans. Power Syst. **30**(3), 1270–1278 (2015)
25. M. Soroush, J.D. Fuller, Accuracies of optimal transmission switching heuristics based on dcopf and acopf. IEEE Trans. Power Syst. **29**(2), 924–932 (2014)